# Maintainer: Robin Candau <antiz@archlinux.org>
# Contributor: robertfoster

pkgname=distrobox
pkgver=1.6.0.1
pkgrel=1
pkgdesc="Use any linux distribution inside your terminal."
arch=('any')
url="https://distrobox.it"
license=('GPL3')
optdepends=('docker: Docker support'
	    'podman: Podman support')
source=("${pkgname}-${pkgver}.tar.gz::https://github.com/89luca89/distrobox/archive/${pkgver}.tar.gz")
sha256sums=('d6b1330b56f6a1bf844c26a27d87f39efd8ae088ed3063f6513d48cf9c18f57e')

package() {
	cd "${pkgname}-${pkgver}"
	./install -P "${pkgdir}/usr"
	find docs -name "*.md" -exec install -Dm 644 -t "${pkgdir}/usr/share/doc/${pkgname}" {} +
}
